Ventana Ground Fault Sampling Scheme

P. McGill

v 1.0.0, 9Oct23

Create an exponentially decaying signal with 5% noise to simulate a ground-fault channel with capacitance. Set the final value of the fault to 20%

```
data = Table[20 + 80 E^(-t/100) + RandomReal[{-2.5, 2.5}], {t, 0, 599}];
```

```
dataPlot = ListPlot[data, DataRange → {0, 30}, PlotRange → All,
  Frame → True, FrameLabel → {{"Fault Reading (percent)", ""},
    {"Time (seconds)", "Fault Reading vs Time"}}, GridLines → Automatic]
```



Find the average of each second of data (20 samples).

```
avgs = Mean /@ Partition[data, 20]
```

```
{92.7708, 79.5429, 69.287, 60.0282, 52.7387, 46.669, 41.6289,
 37.6796, 34.2063, 31.6915, 29.6083, 28.4787, 26.6161, 25.4076,
 24.3906, 23.599, 23.1816, 22.4988, 21.9967, 21.0271, 21.2941, 21.1319,
 20.3384, 20.5381, 21.6124, 20.886, 20.7395, 20.3556, 20.2174, 20.1121}
```

Find the difference of the averages from each second to the next.

```
diffs = Differences[avgs]
```

```
{-13.2278, -10.2559, -9.25881, -7.28953, -6.06971, -5.04001, -3.94937, -3.47331,
 -2.51476, -2.08322, -1.12957, -1.86258, -1.20858, -1.01696, -0.791573,
 -0.417435, -0.682766, -0.502157, -0.969514, 0.266955, -0.162204, -0.793477,
 0.19964, 1.07439, -0.726465, -0.146501, -0.383856, -0.138188, -0.105301}
```

Find which differences are less than 1%.

```
tests = LessThan[1] /@ Abs[diffs]
```

```
{False, False, False, False, False, False, False, False,
 False, False, False, False, False, False, True, True, True, True,
 True, True, True, True, True, False, True, True, True, True, True}
```

Find the time of the first difference that's less than 1%.

```
settledAvgPos = FirstPosition[tests, True] // First
```

```
15
```

Plot the results with the "settled" samples in green.

```
avgPlotBad = ListPlot[avgs[[1 ;; settledAvgPos - 1]], PlotRange → All, PlotStyle → Red];
avgPlotGood = ListPlot[avgs[[settledAvgPos ;; Length[avgs]]],
    DataRange → {settledAvgPos, Length@avgs}, PlotStyle → Green];
```

```
Show[{dataPlot, avgPlotBad, avgPlotGood}]
```



Try this scheme with a shorter time constant of 1 second.

```
data2 = Table[20 + 80 E^(-t/20) + RandomReal[{-2.5, 2.5}], {t, 0, 599}];
dataPlot2 = ListPlot[data2, DataRange → {0, 30}, PlotRange → All,
    Frame → True, FrameLabel → {{"Fault Reading (percent)", ""},
      {"Time (seconds)", "Fault Reading vs Time"}}, GridLines → Automatic];
avgs2 = Mean /@ Partition[data2, 20];
diffs2 = Differences[avgs2];
tests2 = LessThan[1] /@ Abs[diffs2];
settledAvgPos2 = FirstPosition[tests2, True] // First;
avgPlotBad2 =
  ListPlot[avgs2[[1 ;; settledAvgPos2 - 1]], PlotRange → All, PlotStyle → Red];
avgPlotGood2 = ListPlot[avgs2[[settledAvgPos2 ;; Length[avgs2]]],
    DataRange → {settledAvgPos2, Length@avgs2}, PlotStyle → Green];
Show[{dataPlot2, avgPlotBad2, avgPlotGood2}]
```